

[Description](#)  
[Options](#)[Quick start](#)  
[Remarks and examples](#)[Menu](#)  
[Also see](#)[Syntax](#)

## Description

A range and point plot has three  $y$  variables, such as high, low, and closing daily stock prices, or mean values and the corresponding upper and lower bounds of the 95% confidence intervals.

`twoway rpcap` plots a range and a point, using a capped spike to connect the high and low values and a marker for the point on each capped spike.

## Quick start

Specify a range plot of  $y_1$  and  $y_2$  with capped spikes and markers for  $y_3$

```
twoway rpcap y1 y2 y3 x
```

Specify a horizontal range and point plot with capped spikes

```
twoway rpcap y1 y2 y3 x, horizontal
```

Specify maroon capped spikes

```
twoway rpcap y1 y2 y3 x, lcolor(maroon)
```

Specify maroon markers

```
twoway rpcap y1 y2 y3 x, mcolor(maroon)
```

Specify an overlaid line plot

```
twoway rpcap y1 y2 y3 x || line y4 x, sort
```

Specify labels for the  $y$  axis

```
twoway rpcap y1 y2 y3 x, ylabel(100 200 300 400)
```

## Menu

Graphics > Two-way graph (scatter, line, etc.)

## Syntax

```
twoway rpcap y1var y2var y3var xvar [if] [in] [, options]
```

<i>options</i>	Description
<code>vertical</code>	vertical spikes; the default
<code>horizontal</code>	horizontal spikes
<i>marker_options</i>	change look of markers (color, size, etc.)
<i>line_options</i>	change look of spike and cap lines
<i>colorvar_options</i>	change color of spike and cap lines based on values of a variable
<i>axis_choice_options</i>	associate plot with alternative axis
<i>twoway_options</i>	titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc.

All explicit options are *rightmost*, except `vertical` and `horizontal`, which are *unique*; see [G-4] **Concept: repeated options**.

## Options

`vertical` and `horizontal` specify whether the *y1* and *y2* values are to be presented vertically (the default) or horizontally.

In the default `vertical` case, *y1var* and *y2var* record the minimum and maximum (or maximum and minimum) *y* values to be graphed against each *xvar* value.

If `horizontal` is specified, the values recorded in *y1var* and *y2var* are plotted in the *x* direction, and *xvar* is treated as the *y* value.

*marker\_options* specify how the markers look, including shape, size, color, and outline; see [G-3] **marker\_options**. The `msize()` option sets the cap width and the size of the marker symbol.

*line\_options* specify the look of the lines used to draw the spikes and their caps, including pattern, width, and color; see [G-3] **line\_options**.

*colorvar\_options* specify that the color of the lines used to draw the spikes and their caps be determined by the levels of the numeric variable *colorvar*; see [G-3] **colorvar\_options**.

*axis\_choice\_options* associate the plot with a particular *y* or *x* axis on the graph; see [G-3] **axis\_choice\_options**.

*twoway\_options* are a set of common options supported by all `twoway` graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over `by()` groups, and change some advanced settings. See [G-3] **twoway\_options**.

## Remarks and examples

Remarks are presented under the following headings:

*Typical use*

*Advanced use*

### Typical use


We have daily data recording the values for the S&P 500 in 2001:

```
. use https://www.stata-press.com/data/r19/sp500
(S&P 500)
. list date high low close in 1/5
```

	date	high	low	close
1.	02jan2001	1320.28	1276.05	1283.27
2.	03jan2001	1347.76	1274.62	1347.56
3.	04jan2001	1350.24	1329.14	1333.34
4.	05jan2001	1334.77	1294.95	1298.35
5.	08jan2001	1298.35	1276.29	1295.86

We will use the first 57 observations from these data and create a range plot for high and low prices with markers for the closing prices:


```
. twoway rpcap high low close date in 1/57
```



## Advanced use

`twoway rpcap` can be usefully combined with other `twoway` plottypes (see [G-2] [graph twoway](#)). For example, we add a line plot for the closing price to the graph above, and we change the color of the spikes with `lcolor(gs11)` to give the line plot more prominence.

```
. twoway rpcap high low close date, lcolor(gs11) ||
  line close date || in 1/57
  , legend(label(1 "High/Low/Closing price"))
```



## Also see

- [G-2] [graph twoway rarea](#) — Range plot with area shading
- [G-2] [graph twoway rbar](#) — Range plot with bars
- [G-2] [graph twoway rcap](#) — Range plot with capped spikes
- [G-2] [graph twoway rcapsym](#) — Range plot with spikes capped with marker symbols
- [G-2] [graph twoway rconnected](#) — Range plot with connected lines
- [G-2] [graph twoway rline](#) — Range plot with lines
- [G-2] [graph twoway rscatter](#) — Range plot with markers
- [G-2] [graph twoway rspike](#) — Range and point plot with spikes
- [G-2] [graph twoway rspike](#) — Range plot with spikes
- [G-2] [graph twoway spike](#) — Two-way spike plots

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.

For suggested citations, see the FAQ on [citing Stata documentation](#).

